' Update Class Info 2- © 1993-2002 Aiman Hanna

' This program is similar to "Update Class Info 1", except for that the

' input file includes numbers instead of strings. In this case the read

' and write operations will be performed in a slightly different way.

' The input file in this program is assumed to have a student ID followed

' by 3 marks for three assignments. the numbers are separated by comas.

' For simplicity we assume that marks are only allowed as integers (no 8.5, 6.8, etc.).

Option Base 1

Option Explicit

Const maxstudnum As Integer = 30

Dim rowsfilled As Integer

Dim clarr(maxstudnum, 4) As Long    ' Array of 30 rows and 4 columns

Private Sub form_load()

    ' Once the form is loaded, read the input file into

    ' the clarr array

    Dim id As Long, asg1 As Long, asg2 As Long, asg3 As Long

    Dim filename As String

    Dim spacepos As Integer

    Dim i As Integer, j As Integer

    filename = "C:\comp201\c201_assg_info.dat"

    ' Open the file for input (to read from)

    Open filename For Input As #1

    ' Read from the file until you reach its end. Insert all information

    ' into the array

    Do Until EOF(1)

        i = i + 1

        Input #1, id, asg1, asg2, asg3

        ' Now all fields are obtained for one line. Enter them into the array

        clarr(i, 1) = id

        clarr(i, 2) = asg1

        clarr(i, 3) = asg2

        clarr(i, 4) = asg3

    Loop

   ' Notice that the value of i indicates how many rows of the array are actually

   ' filled. We must keep track of this value

   rowsfilled = i

    ' At this point the file is totally read, close it

    Close #1

End Sub

Private Sub cmdRecordEntry_Click()

    Dim id As Long, asg1 As Long, asg2 As Long, asg3 As Long

    ' Read the entered information and add an entry to the array.

    ' Notice that the program should perform a mass of validations here, however,

    ' to keep this program as simple as possible, it is assumed that the user will

    ' enter all needed, and correct, information. Based on this assumption, it is

    ' normal that the program misbehaves if the user misbehaves!

    id = cmbID.Text

    asg1 = txtAsg1.Text

    asg2 = txtAsg2.Text

    asg3 = txtAsg3.Text

    ' Add the entry at the first empty row in the array

    rowsfilled = rowsfilled + 1

    clarr(rowsfilled, 1) = id

    clarr(rowsfilled, 2) = asg1      ' You could simply write: clarr(rowsfilled, 2) = txtAsg1.Text

    clarr(rowsfilled, 3) = asg2

    clarr(rowsfilled, 4) = asg3

    ' Notice that we only added the entry to the array. Nothing is done to save this

    ' entry to the file yet. This will occur when you click on the Save button.

End Sub

Private Sub cmdShowClassInfo_Click()

    Dim i As Integer, j As Integer

    lstClassInfo.Clear

    cmbID.Clear

    For i = LBound(clarr) To rowsfilled

            lstClassInfo.AddItem clarr(i, 1) & "  " & clarr(i, 2) _

                                 & "  " & clarr(i, 3) & "  " & clarr(i, 4)

            ' insert the ids into the combo box

            cmbID.AddItem clarr(i, 1)

    Next i

End Sub

Private Sub cmdSave_Click()

    ' To save the updates, we open the file for writing and then copy the array to

    ' the file. The function will confirm first before updating the file

    Dim ans As Integer

    Dim i As Integer, j As Integer

    Dim filename As String

    ans = MsgBox("Are you sure you want to save your updates?", vbYesNo, "Confirm Saving Request")

    If ans = vbNo Then

        Exit Sub

    Else

        ' open the file for writing

        filename = "C:\comp201\c201_assg_info.dat"

        Open filename For Output As #1

        For i = LBound(clarr) To rowsfilled

            Write #1, Val(clarr(i, 1)), Val(clarr(i, 2)), _

                      Val(clarr(i, 3)), Val(clarr(i, 4))

        Next i

        ' File has been updated. Close it

        Close #1

        ' Clear the text boxes and the combo box

        cmbID.Clear

        txtAsg1.Text = ""

        txtAsg2.Text = ""

        txtAsg3.Text = ""

    End If

End Sub

Private Sub cmdExit_Click()

    End

End Sub

